
NFORMATIKA berez, edo
batzuen ustez behintzat, zien-
tzi arloko jendeari zuzendua
baldin badago ere, hori ez
dela guztiz egia esan behar. Egia da, zientzi
arlokoa ez denari gehiago kostatzen zaiola
programatzen hastea. Baina bestalde, egia da

informatikaren esparrua ez dela programatzaileen
mundura mugatzen ere; erabiltzaileek ere badutela
ezinbesteko garrantzia honetan alegia.

Jadanik ohartuko zinen, irakurle, programatu nahi
dutenei edo Basic lengoaiaren trikimailuak hobeto
ezagutu nahi dituztenei zuzendutako artikuluak idazten
direla atal honetan. Hori da gure asmoa behintzat.

Asmo hau aurrera eramateko zenbait gai eta joko
desberdin landu dugu aldizkari honetan; zerbait
konkretua eskainiz gauzak hobeto ikasten direla bait
deritzogu.

Has gaitezen, sarrera hau egin ondoren, ale honetan
aurkezten dugunaren berri ematen. Azken artikuluak

II
matematika-arlokoak izan dira, zati-
kien inguruko zenbait programa eta
zenbakiak memorizatzeko jokoa hain
zuzen ere. Ale honetan berriz, izen-

buruak ongi adierazten duenez, SINONIMOAK eta
ANTONIMOAK lantzeko programa bat eskainiko
dugu; hizkuntz arloko programa bat. Baina, pro-
gramari bide eman aurretik, gogora ditzagun kontzeptu
hauen esanahiak zeintzuk diren.

SINONIMOA
Zentzu bera, edo antzekoa, eta forma desberdinak

dituzten hitzak dira sinonimoak.
Bapo eta ederki hitzak elkarren sinonimoak dira,

adibidez.

ANTONIMOA
Hitz baten aurkakoari deritzo antonimo.
Sartu eta atera hitzak elkarren antonimoak dira,

adibidez.

Etxeko ordenadorea

SINONIMOAK
ETA  ANTONIMOAK

Pili Lizaso
eta

Eustakio Arrojeria

10 REM SINONIMOAK ETA ANTONIMOAK
20 DIM TAU$(8,3),AUK$(16),KONT(8,2),H(40),HITZAK(8)
30 KEY OFF:CLS:SCREEN 2:SCREEN 0:COLOR 2,0
40 RANDOMIZE TIMER
50 FOR I=1 TO 40:H(I)=I:NEXT I:KOP=40
60 FOR I=1 TO 8
70      ALE=INT(RND*KOP)+1:HITZAK(I)=H(ALE)
80      FOR J=ALE TO KOP-1:H(J)=H(J+1):NEXT J
90      KOP=KOP-1
100 NEXT I
110 K=-1
120 FOR I=1 TO 8
130 RESTORE 1000
140 FOR J=1 TO HITZAK(I):READTAU$(I,1),TAU$(I,2),TAU$(I,3):NEXT J
150 K=K+2:AUK$(K)=TAU$(I,2):AUK$(K+1)=TAU$(I,3)
160 NEXT I
170 AL$=”B”
180 WHILE AL$=”B”
190   AL$=”E”
200   FOR I=1 TO 15
210       IF AUK$(I)>AUK$(I+1) THEN SWAP AUK$(I),AUK$(I+1):AL$=”B”

220   NEXT I
230 WEND
240 COLOR 10:LOCATE 1,27:PRINT “SINONIMOAK ETA ANTONIMOAK”
250 COLOR 11:LOCATE 2,27:PRINT “=========================”
260 COLOR 10:LOCATE 4,5:PRINT “HITZAK SINONIMOAK ANTONIMOAK”
270 COLOR 11:LOCATE 5,5:PRINT “————                        ———————               ———————”
280 COLOR 3
290 FOR I=1 TO 8
300     LOCATE 7+(I-1)*2,5:PRINT TAU$(I,1):LOCATE 7+(I-1)*2,30:PRINT STRING$(20,45)
310     LOCATE 7+(I-1)*2,55:PRINT STRING$(20,45)
320 NEXT I
330 COLOR 11:LOCATE 24,15:PRINT CHR$(24);:COLOR 3:PRINT “: gora “;:COLOR 11
340 PRINT CHR$(25);:COLOR 3:PRINT “: behera “;:COLOR 11:PRINT CHR$(26);” “;CHR$(27);
350 COLOR 3:PRINT “: emaitza hautatzeko”;
360 LOCATE 25,15:COLOR 11:PRINT “zuriune-barra”;:COLOR 3:PRINT “: hurrengo zutabera “;
370 COLOR 11:PRINT “<RETURN>”;:COLOR 3:PRINT “: amaitzeko”;
380 Z=0:L=1
390 FOR I=1 TO 8
400 KONT(I,1)=0:KONT(I,2)=0
410 NEXT I
420 TE$=INKEY$:WHILE TE$<>””:TE$=INKEY$:WEND:TE$=”A”
430 LOCATE 7,5:COLOR 11:PRINT TAU$(1,1)
440 WHILE ASC(TE$)<>13
450  WHILE ASC(TE$)<>13 AND ASC(RIGHT$(TE$,1))<>72 AND ASC(RIGHT$(TE$,1))<>80 

AND ASC(RIGHT$(TE$,1))<>75 AND ASC(RIGHT$(TE$,1))<>77 AND 
ASC(RIGHT$(TE$,1))<>32

460     TE$=INKEY$:IF TE$=”” THEN TE$=”A”
470   WEND
480   A$=RIGHT$(TE$,1)
490   IF A$<>”H” THEN 540
500      IF L<2 OR Z>0 THEN 530
510         LOCATE 7+(L-1)*2,5:COLOR 3:PRINT TAU$(L,1):L=L-1
520         LOCATE 7+(L-1)*2,5:COLOR 11:PRINT TAU$(L,1):GOTO 540
530      BEEP
540   IF A$<>”P” THEN 590
550      IF L>7 OR Z>0 THEN 580
560         LOCATE 7+(L-1)*2,5:COLOR 3:PRINT TAU$(L,1):L=L+1
570         LOCATE 7+(L-1)*2,5:COLOR 11:PRINT TAU$(L,1):GOTO 590
580      BEEP
590   IF A$<>”M” THEN 650
600      IF Z=0 OR KONT(L,Z)>15 THEN 640
610         LOCATE 7+(L-1)*2,5+Z*25:PRINT SPC(25):KONT(L,Z)=KONT(L,Z)+1
620         COLOR 11:LOCATE 7+(L-1)*2,40+(Z-1)*25-INT(LEN(AUK$(KONT(L,Z)))/2)
630         PRINT AUK$(KONT(L,Z)):GOTO 650
640      BEEP
650   IF A$<>”K” THEN 710
660      IF Z=0 OR KONT(L,Z)<2 THEN 700
670         LOCATE 7+(L-1)*2,5+Z*25:PRINT SPC(25):KONT(L,Z)=KONT(L,Z)-1:COLOR 11
680         LOCATE 7+(L-1)*2,40+(Z-1)*25-INT(LEN(AUK$(KONT(L,Z)))/2)
690         PRINT AUK$(KONT(L,Z)):GOTO 710
700      BEEP
710 IF A$<>” “ THEN 770
720 COLOR 3:IF Z=0 THEN LOCATE 7+(L-1)*2,5:PRINT TAU$(L,1)

ELSE LOCATE 7+(L-1)*2,40+(Z-1)*25-INT(LEN(AUK$(KONT(L,Z)))/2): 
PRINT AUK$(KONT(L,Z))

730      IF Z=2 THEN Z=0:COLOR 11:LOCATE 7+(L-1)*2,5:PRINT TAU$(L,1):GOTO 770
740      Z=Z+1:LOCATE 7+(L-1)*2,30+(Z-1)*25:PRINT SPC(20):COLOR 11
750      IF KONT(L,Z)=0 THEN KONT(L,Z)=1
760      LOCATE 7+(L-1)*2,40+(Z-1)*25-INT(LEN(AUK$(KONT(L,Z)))/2):PRINT AUK$(KONT(L,Z))
770   IF ASC(TE$)<>13 THEN TE$=”A”
780 WEND
790 ASMAT$=”E”
800 FOR I=1 TO 8
810     FOR J=1 TO 2
820         IF TAU$(I,J+1)=AUK$(KONT(I,J)) THEN COLOR 11

ELSE COLOR 12:ASMAT$=”F”
830 LOCATE 7+(I-1)*2,30+(J-1)*25:PRINT SPC(20)

Programaren azalpena:

20-100: 1etik 40ra bitarteko 8 zenbaki aleatorio
desberdin hautatzen dira eta HITZAK
taulan gordetzen.

110-160: Aleatorioki ateratako zenbakiei dagozkien
hitzak (hirukoteak) irakurri eta sinonimo
eta antonimoak AUK$ taulan gordetzen
dira.

170-230: AUK$ taulako balioak ordena alfabetiko
gorakorrean ordenatzen dira.

240-370: Hautatutako hitzen aurkezpena eta erabil
daitezkeen teklen azalpena:

←, → teklak: aukera daitezkeen 16
balioen artean bat hautatzeko
↑ , ↓ teklak: hirukote batetik bestera pa-
satzeko 
zuriune-barra: zutabe batetik bestera pa-
satzeko
<RETURN>: programa amaitzeko

380-780: Sakatzen diren teklen kontrola egiten da.
Higidura-teklak bakarrik onartzen dira eta
prozesua <RETURN> sakatutakoan

amaitzen da. ASCII koderik ez duen te-
klaren bat sakatzen denean, higidura-
teklak adibidez, INKEY$ funtzioak bi
karakterez osaturiko katea itzultzen du.
Katea honetako lehen karakterea nulua
da, 0 ASCII kodea duena; horregatik
erabiltzen da hain zuzen ere RIGHT$
funtzioa, kateatik bigarren karakterea
bakarrik hartzeko.
Bestalde, egiten diren aukerak
KONT(L,Z)n, non L-k lerroa eta Z-k
zutabea adierazten duten, gordetzen dira.
1etik 16ra bitarteko zenbakiak izango dira
taula honetan gordeko diren balioak.

790-910: Erabiltzaileak egindako aukeren baiezta-
pena burutzen da zati honetan; asma-
tutakoak urdinez eta huts egindakoak
gorriz agertuz.

1000-1190: Programan erabiltzen diren datuak. Datu
hauek erabiltzailearen nahiaren arabera
alda daitezke.
Kasu honetan 40 hirukote hartu badira ere
gehiago edo gutxiago ere izan daitezke;
hori bai, 20 eta 50 lerroetan 40 zenba-
kiaren ordez dagokiona ipini beharko
litzateke.   y

840         LOCATE 7+(I-1)*2,40+(J-1)*25-INT(LEN(TAU$(I,J+1))/2)
850         PRINT TAU$(I,J+1)
860 NEXT J
870 NEXT I
880 IF ASMAT$=”F” THEN 910
890    LOCATE 24,1:PRINT SPC(79);:LOCATE 25,1:PRINT SPC(79);
900    LOCATE 25,23:COLOR 11:PRINT “OSO ONGI!!! Guztiak asmatu dituzu.”;: LOCATE 1,1
910 END
920 REM
1000 DATA “aberats”,”dirudun”,”behartsu”,”abiadura”,”ziztu”,”geldotasun”
1010 DATA “triste”,”goibel”,”alai”,”arbuiagarri”,”zirtzil”,”estimagarri”
1020 DATA “menpeko”,”esaneko”,”arduradun”,”asperkunde”,”mendeku”,”barkamen”
1030 DATA “azkargarri”,”indargarri”,”lasaigarri”,”bakan”,”harrigarri”,”arrunt”
1040 DATA “bapo”,”ederki”,”txarto”,”beldur”,”izu”,”adore”
1050 DATA “eskuzabal”,”emankor”,”berekoi”,”pottolo”,”gizen”,”argal”
1060 DATA “bizi”,”azkar”,”astiro”,”damu”,”pena”,”alaitasun”
1070 DATA “aski”,”sobera”,”urri”,”baldar”,”trakets”,”trebe”
1080 DATA “ebakidura”,”etena”,”lotura”,”egur”,”jipoi”,”laztan”
1090 DATA “elkartu”,”pilatu”,”banatu”,”lasaitu”,”baretu”,”aztoratu”
1100 DATA “etekin”,”mozkin”,”galera”,”fidagarri”,”sinesgarri”,”sineskaitz”
1110 DATA “gaizkor”,”erikor”,”osasuntsu”,”gogaikarri”,”nazkagarri”,”atsedengarri”
1120 DATA “gupidatsu”,”bihozbera”,”kupidagabe”,”hasiera”,”iturburu”,”amaiera”
1130 DATA “hoben”,”akats”,”birtute”,”ikara”,”beldur”,”bare”
1140 DATA “iruzur”,”tranpa”,”egia”,”jabe”,”ugazaba”,”morroi”
1150 DATA “jaulkigarri”,”kontagarri”,”kontaezin”,”jubilazio”,”erretiro”,”lansasoi”
1160 DATA “kautu”,”zentzudun”,”zentzugabe”,”koipetsu”,”eztitsu”,”zakar”
1170 DATA “lagun”,”adiskide”,”etsai”,”leial”,”zintzo”,”maltzur”
1180 DATA “mazkildura”,”zikin”,”garbidura”,”nekazari”,”laborari”,”kaletar”
1190 DATA “oihan”,”baso”,”basamortu”,”xume”,”umil”,”harro”


